

#Mem Addr	#Machine Code	#Branch/Call Target	#Mnemonic	#Comment
				Assemble Instructions
				Monarch2090
				##### Register use #####
		#define	LO r4	Walze links oben
		#define	LU r3	Walze links unten
		#define	MI r2	Walze in der Mitte
		#define	RO r0	Walze rechts oben
		#define	RU r1	Walze rechts unten
		#define	GELD_PF_EINER rC	Münzspeicher 1 Pf (immer 0)
		#define	GELD_PF_ZEHNER rD	Münzspeicher 10 Pf
		#define	GELD_DM_EINER rE	Münzspeicher 1 DM
		#define	GELD_DM_ZEHNER rF	Münzspeicher 10 DM
		#define	S_EINER rD	Sonderspiele Einer (im Speicherregister)
		#define	S_ZEHNER rE	Sonderspiele Zehner (im Speicherregister)
		#define	S_HUNDERTER rF	Sonderspiele Hunderter (im SR, immer 0)
		#define	DELTA_C r4	Warteschleife setzt auf 0
		#define	DELTA_D r5	Gewinn/Abzug Geld/Sonderspiele
		#define	DELTA_E r6	Gewinn/Abzug Geld/Sonderspiele
		#define	DELTA_F r7	Gewinn/Abzug Geld/Sonderspiele
		#define	ZÄHLER r8	für Warteschleife (vorher 0, nachher 0)
		#define	ZAHL09 rD	Zufallszahl über RND
		#define	TASTE r9	Rote Taste zum Nachstart
		#define	STATUS rA	Wenn Sonderspiele, dann 2, sonst 0
		#define	CHANCE1 r8	Symbol links und rechts gleich I
		#define	CHANCE2 r9	Symbol links und rechts gleich II
		#define	GEWINNSYMBOL rB	erzieltes Gewinnsymbol
				##### Program #####
0x00	F08	start0	CLEAR	
0x01	F4C		DISP #4,GELD_PF_EINER	
0x02	FFE		KIN GELD_DM_EINER	Geld einwerfen
0x03	F04		DZHX	
0x04	F4C	start	DISP #4,GELD_PF_EINER	Münzspeicher anzeigen
0x05	1E5		MOVI #E,DELTA_D	-20 Pf
0x06	BE6		CALL DeltaSub	
0x07	9FF		CMPI #F,GELD_DM_ZEHNER	Münzspeicher unter 0?
0x08	E00		BRZ start0	dann zurück zum Anfang
0x09	BEF		CALL Warteschleife	
0x0A	FEA		DOT STATUS	Keine Sonderspiele?
0x0B	E10		BRZ StartSpiel	dann direkt Spiel starten
0x0C	F0E		EXRM	
0x0D	F2D		DISP #2,S_EINER	Sonderspiele-Zähler anzeigen
0x0E	BEF		CALL Warteschleife	
0x0F	F0E		EXRM	
0x10	0D5	StartSpiel	MOV GELD_PF_ZEHNER,DELTA_D	Münzspeicher sichern
0x11	0E6		MOV GELD_DM_EINER,DELTA_E	
0x12	0F7		MOV GELD_DM_ZEHNER,DELTA_F	
0x13	F50		DISP #5,RO	Walzen darstellen

#Mem Addr	#Machine Code	#Branch/Call Target	#Mnemonic	#Comment
				Assemble Instructions
0x14	BF6	stopL	CALL HoleZahl	Linke Walze ermitteln
0x15	1F4		MOVI #F,LO	König
0x16	1D3		MOVI #D,LU	2 DM
0x17	98D		CMPI #8,ZAHL09	
0x18	D2D		BRC L_fertig	
0x19	1E4		MOVI #E,LO	Krone
0x1A	193		MOVI #9,LU	1 DM
0x1B	97D		CMPI #7,ZAHL09	
0x1C	D2D		BRC L_fertig	
0x1D	1B4		MOVI #B,LO	1,40 DM (1 DM LU schon gesetzt)
0x1E	96D		CMPI #6,ZAHL09	
0x1F	D2D		BRC L_fertig	
0x20	184		MOVI #8,LO	80 Pf (1 DM LU schon gesetzt)
0x21	95D		CMPI #5,ZAHL09	
0x22	D2D		BRC L_fertig	
0x23	143		MOVI #4,LU	40 Pf (80 Pf LO schon gesetzt)
0x24	94D		CMPI #4,ZAHL09	
0x25	D2D		BRC L_fertig	
0x26	1B4		MOVI #B,LO	1,40 DM (40 Pf RU schon gesetzt)
0x27	93D		CMPI #3,ZAHL09	
0x28	D2D		BRC L_fertig	
0x29	1A4		MOVI #A,LO	1,20 DM (40 Pf RU schon gesetzt) (2x)
0x2A	91D		CMPI #1,ZAHL09	
0x2B	D2D		BRC L_fertig	
0x2C	1C4		MOVI #C,LO	1,60 DM (40 Pf RU schon gesetzt) (2x)
0x2D	BEF	L_fertig	CALL Warteschleife	
0x2E	FD9		DIN TASTE	
0x2F	588		ADDI #8,ZÄHLER	nur einmal Nachstarten ermöglichen
0x30	D33		BRC stopR	
0x31	909		CMPI #0,TASTE	Taste gedrückt?
0x32	D14		BRC stopL	Dann linke Walze nochmal starten
0x33	BF6	stopR	CALL HoleZahl	Rechte Walze ermitteln
0x34	1F0		MOVI #F,RO	König
0x35	1D1		MOVI #D,RU	2 DM
0x36	98D		CMPI #8,ZAHL09	
0x37	D49		BRC R_fertig	
0x38	1E0		MOVI #E,RO	Krone
0x39	191		MOVI #9,RU	1 DM
0x3A	97D		CMPI #7,ZAHL09	
0x3B	D49		BRC R_fertig	
0x3C	180		MOVI #8,RO	80 Pf (1 DM RU schon gesetzt)
0x3D	96D		CMPI #6,ZAHL09	
0x3E	D49		BRC R_fertig	
0x3F	1B1		MOVI #B,RU	1,40 DM (80 Pf RO schon gesetzt)
0x40	95D		CMPI #5,ZAHL09	
0x41	D49		BRC R_fertig	
0x42	141		MOVI #4,RU	40 Pf (80 Pf RO schon gesetzt) (2x)
0x43	93D		CMPI #3,ZAHL09	
0x44	D49		BRC R_fertig	
0x45	1A0		MOVI #A,RO	1,20 DM (40 Pf RU schon gesetzt) (2x)
0x46	91D		CMPI #1,ZAHL09	
0x47	D49		BRC R_fertig	
0x48	1C0		MOVI #C,RO	1,60 DM (40 Pf RU schon gesetzt) (2x)
0x49	BEF	R_fertig	CALL Warteschleife	
0x4A	BF6	stopM	CALL HoleZahl	Mittlere Walze ermitteln
0x4B	1F2		MOVI #F,MI	König
0x4C	98D		CMPI #8,ZAHL09	
0x4D	D5B		BRC M_fertig	
0x4E	1C2		MOVI #C,MI	1,60 DM

#Mem Addr	#Machine Code	#Branch/Call Target	#Mnemonic	#Comments
				Assemble Instructions
0x4F	97D		CMPI #7,ZAHL09	
0x50	D5B		BRC M_fertig	
0x51	142		MOVI #4,MI	40 Pf (2x)
0x52	95D		CMPI #5,ZAHL09	
0x53	D5B		BRC M_fertig	
0x54	182		MOVI #8,MI	80 Pf (2x)
0x55	93D		CMPI #3,ZAHL09	
0x56	D5B		BRC M_fertig	
0x57	1A2		MOVI #A,MI	1,20 DM (2x)
0x58	91D		CMPI #1,ZAHL09	
0x59	D5B		BRC M_fertig	
0x5A	1D2		MOVI #D,MI	2 DM (2x)
0x5B	BEF	M_fertig	CALL Warteschleife	
0x5C	05D		MOV DELTA_D,GELD_PF_ZEHNER	Münzspeicher wiederherstellen
0x5D	06E		MOV DELTA_E,GELD_DM_EINER	
0x5E	07F		MOV DELTA_F,GELD_DM_ZEHNER	
				Gewinnabtastung
0x5F	109		MOVI #0,CHANCE2	
0x60	10B		MOVI #0,GEWINNSYMBOL	
0x61	BFA		CALL LöscheDelta	
0x62	840	ChanceA	CMP LO,RO	
0x63	E67		BRZ erste_Chance	
0x64	841		CMP LO,RU	
0x65	E67		BRZ erste_Chance	
0x66	C68		GOTO ChanceB	
0x67	048	erste_Chance	MOV LO,CHANCE1	
0x68	830	ChanceB	CMP LU,RO	
0x69	E6D		BRZ zweite_Chance	
0x6A	831		CMP LU,RU	
0x6B	E6D		BRZ zweite_Chance	
0x6C	C6E		GOTO Mitte	
0x6D	039	zweite_Chance	MOV LU,CHANCE2	
0x6E	9E2	Mitte	CMPI #E,MI	König (Joker) in der Mitte?
0x6F	D79		BRC Joker_ja	
0x70	828	Joker_nein	CMP MI,CHANCE1	Normale Gewinnmöglichkeit
0x71	E73		BRZ gewinn1	
0x72	C74		GOTO test2	
0x73	08B	gewinn1	MOV CHANCE1,GEWINNSYMBOL	
0x74	829	test2	CMP MI,CHANCE2	
0x75	E77		BRZ gewinn2	
0x76	C80		GOTO symbol	
0x77	09B	gewinn2	MOV CHANCE2,GEWINNSYMBOL	
0x78	C80		GOTO symbol	
0x79	12B	Joker_ja	MOVI #2,GEWINNSYMBOL	Mindestens 20 Pf
0x7A	88B		CMP CHANCE1,GEWINNSYMBOL	Höchste Chance = Gewinn
0x7B	D7D		BRC aaa	
0x7C	08B		MOV CHANCE1,GEWINNSYMBOL	besserer Gewinn
0x7D	89B	aaa	CMP CHANCE2,GEWINNSYMBOL	
0x7E	D80		BRC symbol	
0x7F	09B		MOV CHANCE2,GEWINNSYMBOL	bester Gewinn
0x80	99B	symbol	CMPI #9,GEWINNSYMBOL	1 DM?
0x81	E84		BRZ G100	
0x82	D85		BRC Gewinn120	
0x83	C94		GOTO GewinnBis80Pf	

#Mem Addr	#Machine Code	#Branch/Call Target	#Mnemonic	#Comment
				Assemble Instructions
0x84	1A5	G100	MOVI #A,DELTA_D	Hex 10
0x85	9AB	Gewinn120	CMPI #A,GEWINNSYMBOL	1,20 DM?
0x86	E88		BRZ G120	
0x87	C89		GOTO Gewinn140	
0x88	1C5	G120	MOVI #C,DELTA_D	Hex 12
0x89	9BB	Gewinn140	CMPI #B,GEWINNSYMBOL	1,40 DM?
0x8A	E8C		BRZ G140	
0x8B	C8D		GOTO Gewinn160	
0x8C	1E5	G140	MOVI #E,DELTA_D	Hex 14
0x8D	9CB	Gewinn160	CMPI #C,GEWINNSYMBOL	1,60 DM?
0x8E	E91		BRZ G160	
0x8F	D9E		BRC Erhöhung	Mehr als 1,60 DM, dann 2 DM geben (Gewinnkombinationen DDD, EFE, FFF)
0x90	C95		GOTO SoSpTest	
0x91	105	G160	MOVI #0,DELTA_D	Hex 16
0x92	116		MOVI #1,DELTA_E	Hex 16
0x93	C95		GOTO SoSpTest	
0x94	0B5	GewinnBis80Pf	MOV GEWINNSYMBOL,DELTA_D	00 Pf, 20 Pf, 40 Pf, 80 Pf (Hex = Dez)
0x95	90A	SoSpTest	CMPI #0,STATUS	Laufen gerade keine Sonderspiele?
0x96	EA0		BRZ GongTest	
0x97	90B	Test2DM	CMPI #0,GEWINNSYMBOL	Irgendeinen Gewinn erzielt?
0x98	D9E		BRC Erhöhung	dann Gewinn auf 2 DM erhöhen
0x99	942	TestKaro	CMPI #4,MI	Mitte 40 Pf?
0x9A	E9E		BRZ Erhöhung	dann Gewinn auf 2 DM erhöhen
0x9B	9C2		CMPI #C,MI	Mitte 2 DM oder König?
0x9C	D9E		BRC Erhöhung	dann Gewinn auf 2 DM erhöhen
0x9D	CA6		GOTO GewinnAnzeige	
0x9E	145	Erhöhung	MOVI #4,DELTA_D	2 DM (Hex 20)
0x9F	116		MOVI #1,DELTA_E	
0xA0	906	GongTest	CMPI #0,DELTA_E	
0xA1	DA4		BRC Ton	
0xA2	905		CMPI #0,DELTA_D	Nichts gewonnen?
0xA3	EA6		BRZ GewinnAnzeige	Dann nur anzeigen
0xA4	70F	Ton	SUBI #0,rF	Sound erzeugen (spez. OpCode)
0xA5	000		MOV r0,r0	Sound stoppen (spez. OpCode)
0xA6	F0F	GewinnAnzeige	EXRA	Gewinn anzeigen
0xA7	F3C		DISP #3,GELD_PF_EINER	
0xA8	BEF		CALL Warteschleife	setzt auch R0 auf 0 wg. EXTRA
0xA9	F02		DISOUT	
0xAA	F0F		EXRA	
0xAB	BE8		CALL DeltaAdd	
0xAC	BFA		CALL LöscheDelta	Geldgewinne löschen
0xAD	90A		CMPI #0,STATUS	Laufen keine Sonderspiele?
0xAE	EC7		BRZ SoGewinn	dann EFE oder FFF testen
0xAF	F0E	TableauTest	EXRM	Sonderspiele-Register einblenden
0xB0	1F5		MOVI #F,DELTA_D	vorher -1 Sonderspiel (akt. Spiel)
0xB1	BE6		CALL DeltaSub	
0xB2	BFA		CALL LöscheDelta	

#Mem Addr	#Machine Code	#Branch/Call Target	#Mnemonic	#Comment
				Assemble Instructions
0xB3	7F2		SUBI #F,MI	Tableau testen: König in der Mitte?
0xB4	DC6		BRC EndeTableau	kein König in der Mitte
0xB5	90E	Tableau	CMPI #0,S_ZEHNER	Pos > 16?
0xB6	DC6		BRC EndeTableau	dann keine Tableaugewinne
0xB7	99D		CMPI #9,S_EINER	Pos > 10?
0xB8	DC0		BRC Pos16	dann Pos16 und Pos15 prüfen
0xB9	EBE		BRZ Gewinn10S	Pos = 10? dann 10 Sonderspiele
0xBA	125		MOVI #2,DELTA_D	Pos < 10? dann mind. 2 Sonderspiele
0xBB	90D		CMPI #0,S_EINER	Pos = 1?
0xBC	DC6		BRC EndeTableau	
0xBD	585	Gewinn20S	ADDI #8,DELTA_D	20 Sonderspiele geben:
				2 Sonderspiele aufstocken auf 10
				und zusätzlich...
0xBE	5A5	Gewinn10S	ADDI #A,DELTA_D	10 Sonderspiele geben
0xBF	FB6		ADC DELTA_E	
0xC0	9ED	Pos16	CMPI #E,S_EINER	Pos > 15?
0xC1	DC4		BRC Gewinn50S	Dann 50 Sonderspiele
0xC2	EC5		BRZ Gewinn2S	Pos = 15? dann 2 Sonderspiele
0xC3	CC6		GOTO EndeTableau	
0xC4	136	Gewinn50S	MOVI #3,DELTA_E	50 Sonderspiele (Hex 32)
0xC5	125	Gewinn2S	MOVI #2,DELTA_D	
0xC6	F0E	EndeTableau	EXRM	Sonderspiele-Register ausblenden
0xC7	906	SoGewinn	CMPI #0,DELTA_E	Über Tableau >15 Sonderspiele gewonnen?
0xC8	DD3		BRC SoAnzeige	dann anzeigen, da mehr als über EFE/FFF
0xC9	9A5		CMPI #A,DELTA_D	Genau 10 gewonnen?
0xCA	ED3		BRZ SoAnzeige	dann anzeigen
0xCB	9EB	TestEFE	CMPI #E,GEWINNSYMBOL	Sonderspiele über EFE/FFF gewonnen?
0xCC	DD1		BRC ZehnS	
0xCD	ECF		BRZ FünfS	
0xCE	CD3		GOTO SoAnzeige	
0xCF	155	FünfS	MOVI #5,DELTA_D	5 Sonderspiele geben
0xD0	CD2		GOTO SoGewonnen	
0xD1	1A5	ZehnS	MOVI #A,DELTA_D	10 Sonderspiele geben
0xD2	12A	SoGewonnen	MOVI #2,STATUS	Statusregister auf "Sonderspiele" setzen
0xD3	90A	SoAnzeige	CMPI #0,STATUS	Keine Sonderspiele?
0xD4	EE3		BRZ Aufräumen	dann keine Anzeige und aufräumen
0xD5	F0F		EXRA	Sonderspiele-Gewinn anzeigen
0xD6	F2D		DISP #2,S_EINER	
0xD7	BEF		CALL Warteschleife	
0xD8	F0F		EXRA	
0xD9	F02		DISOUT	
0xDA	F0E		EXRM	Sonderspiele-Register einblenden
0xDB	BE8		CALL DeltaAdd	mögl. Sonderspiele-Gewinn addieren
0xDC	90E		CMPI #0,S_ZEHNER	Sonderspiele-Zähler 0?
0xDD	DE2		BRC SoAnzeigeEnde	
0xDE	90D		CMPI #0,S_EINER	
0xDF	DE2		BRC SoAnzeigeEnde	

[illegible]

[illegible]

[illegible]

[illegible]

[illegible]

[illegible]

[illegible]